
 
 
 
 
 
 
 
 
 
 
 
 

 
 
 
 

 
 

LAMPIRAN A 

FOTO ALAT SISTEM PENGENDALI ON-OFF 

LAMPIRAN B 

PROGRAM PADA PENGONTROL ATMEL 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



 
PROGRAM PADA BAGIAN REMOTE CONTROL 

 
#INCLUDE        "8051.H" 
 
RELAYPTT        .EQU    P2.7 
E_LCD           .EQU    P1.0 
RW_LCD          .EQU    P1.1 
RS_LCD          .EQU    P1.2 
DATALCD         .EQU    P0 
 
SWITCH          .EQU    P2 
X1              .EQU    P2.4 
X2              .EQU    P2.5 
X3              .EQU    P2.6 
Y1              .EQU    P2.0 
Y2              .EQU    P2.1 
Y3              .EQU    P2.2 
Y4              .EQU    P2.3 
 
 
                .ORG    $30 
DATAKEY         .BLOCK  1 
DATAKIRIM       .BLOCK  1 
                .ORG    $0 
                LJMP    MULAI 
 
 
                .ORG    $100 
MULAI:          MOV     SP,#$20 
                MOV     PSW,#0 
                LCALL   DELAYSW 
                CLR     RELAYPTT 
                LCALL   INIT_LCD 
                LCALL   INITSERIAL 
                MOV     DPTR,#TEXT1 
                LCALL   PROC_STRTOLCD 
                LCALL   PROC_LFLCD 
                MOV     DPTR,#TEXT2 
                LCALL   PROC_STRTOLCD 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
 
 
LOOP:           MOV     SP,#$20 
                LCALL   PROC_CLEARLCD 
                MOV     DPTR,#TEXTPERINTAH1 
                LCALL   PROC_STRTOLCD 
                LCALL   PROC_LFLCD 
                MOV     DPTR,#TEXTPERINTAH2 
                LCALL   PROC_STRTOLCD 



ULANGI: 
 
                LCALL   CEKKEYPAD 
                LCALL   KEYPADTOLCD 
 
                SETB    RELAYPTT 
                LCALL   DELAY5X 
 
                MOV     R1,#2 
ULANGKIRIM0: 
                MOV     A,#$55 
                LCALL   SENDCHR 
                MOV     A,#$EE 
                LCALL   SENDCHR 
                MOV     A,#$AA 
                LCALL   SENDCHR 
                 
 
                MOV     R2,#200 
ULANGKIRIM1: 
                MOV     A,DATAKIRIM 
                LCALL   SENDCHR 
                MOV     A,#$0D 
                LCALL   SENDCHR 
                DJNZ    R2,ULANGKIRIM1 
                DJNZ    R1,ULANGKIRIM0 
                 
 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
                CLR     RELAYPTT 
LOOPUL:         LJMP    LOOP 
 
 
CEKKEYPAD:      LCALL   SCANNINGKEYPAD 
                CJNE    A,#$0,CEKKEYPASS 
                LJMP    CEKKEYPAD 
CEKKEYPASS:     RET 
 
 
PROC_HOMELCD: 
                MOV     A,#02H 
                LCALL   WRITE_CTRLLCD 
                RET 
;------------------------------------------------------- 
; ROUTINE LF LCD MENEMPATKAN KURSOR PADA BARIS KE 2 LCD 
;------------------------------------------------------- 
PROC_LFLCD: 
                MOV     A,#0C0H 
                LCALL   WRITE_CTRLLCD 
                RET 
;------------------------------------------------------- 
; ROUTINE CLEAR LCD MENGHAPUS LAYAR LCD 
;------------------------------------------------------- 
PROC_CLEARLCD: 
                MOV     A,#01H 
                LCALL   WRITE_CTRLLCD 



                RET 
 
;------------------------------------------------------- 
; ROUTINE PENULISAN DATA KE LCD 
;------------------------------------------------------- 
WRITE_DATALCD: 
                MOV     DATALCD,A 
 
                CLR     RW_LCD 
                SETB    RS_LCD 
                CLR     E_LCD 
                LCALL   DELAY_LCD 
 
                CLR     RW_LCD 
                SETB    RS_LCD 
                SETB    E_LCD 
                LCALL   DELAY_LCD 
 
                CLR     RW_LCD 
                SETB    RS_LCD 
                CLR     E_LCD 
                NOP 
                SETB    RW_LCD 
                CLR     RS_LCD 
                CLR     E_LCD 
                LCALL   DELAY_LCD 
                RET 
 
;------------------------------------------------------- 
; ROUTINE PENULISAN PERINTAH/COMMAND KE LCD 
;------------------------------------------------------- 
WRITE_CTRLLCD: 
;                           RW  RS  E 
;                           0   1   0 
                MOV     DATALCD,A 
             
                CLR     RW_LCD 
                CLR     RS_LCD 
                CLR     E_LCD 
                LCALL   DELAY_LCD 
 
                CLR     RW_LCD 
                CLR     RS_LCD 
                SETB    E_LCD 
                LCALL   DELAY_LCD 
 
                CLR     RW_LCD 
                CLR     RS_LCD 
                CLR     E_LCD 
                NOP 
                SETB    RW_LCD 
                CLR     RS_LCD 
                CLR     E_LCD 
                LCALL   DELAY_LCD 
             
                RET 
 



;------------------------------------------------------ 
;       Routine Pengiriman String ke Display LCD 
;------------------------------------------------------ 
PROC_STRTOLCD: 
                CLR     A 
                MOVC    A,@A+DPTR 
                CJNE    A,#00H,STRTOLCD 
                RET 
STRTOLCD: 
                LCALL   WRITE_DATALCD 
                INC     DPTR 
                SJMP    PROC_STRTOLCD 
 
 
;------------------------------------------------------- 
; ROUTINE INISIALISASI LCD 
;------------------------------------------------------- 
INIT_LCD: 
                MOV     A,#38H 
                LCALL   WRITE_CTRLLCD 
                MOV     A,#0EH 
                LCALL   WRITE_CTRLLCD 
                MOV     A,#0CH 
                LCALL   WRITE_CTRLLCD 
                MOV     A,#06H 
                LCALL   WRITE_CTRLLCD 
                MOV     A,#01H 
                LCALL   WRITE_CTRLLCD 
                RET 
 
INITSERIAL: 
                MOV     TMOD,#20H 
                MOV     TCON,#41H 
                MOV     TH1,#$D0 
                MOV     SCON,#50H 
                RET 
 
SENDCHR: 
                CLR     TI 
                MOV     SBUF,A 
TXLOOP: 
                JNB     TI,TXLOOP 
                RET 
 
 
ANDF0LCD:       SWAP    A 
AND0FLCD: 
                ANL     A,#$0F 
                ADD     A,#$30 
                LCALL   WRITE_DATALCD 
                RET 
 
SCANNINGKEYPAD: 
KOLOM1:         MOV     DATAKEY,#0 
                CLR     X1 
                SETB    X2 
                SETB    X3 



                MOV     A,SWITCH 
                ANL     A,#0FH 
CTOMBOL1:       CJNE    A,#0EH,CTOMBOL4 
                MOV     DATAKEY,#01H 
                MOV     DATAKIRIM,#$1 
                LJMP    TOLCD 
CTOMBOL4:       CJNE    A,#0DH,CTOMBOL7 
                MOV     DATAKEY,#04H 
                MOV     DATAKIRIM,#$4 
                LJMP    TOLCD 
CTOMBOL7:       CJNE    A,#0BH,CTOMBOLB 
                MOV     DATAKEY,#07H 
                MOV     DATAKIRIM,#$7 
                LJMP    TOLCD 
CTOMBOLB:       CJNE    A,#07H,KOLOM2 
                MOV     DATAKEY,#0BH 
                MOV     DATAKIRIM,#'B' 
                LJMP    TOLCD 
 
KOLOM2:         SETB    X1 
                CLR     X2 
                SETB    X3 
                MOV     A,SWITCH 
                ANL     A,#0FH 
CTOMBOL2:       CJNE    A,#0EH,CTOMBOL5 
                MOV     DATAKEY,#02H 
                MOV     DATAKIRIM,#$2 
                LJMP    TOLCD 
CTOMBOL5:       CJNE    A,#0DH,CTOMBOL8 
                MOV     DATAKEY,#05H 
                MOV     DATAKIRIM,#$5 
                LJMP    TOLCD 
CTOMBOL8:       CJNE    A,#0BH,CTOMBOL0 
                MOV     DATAKEY,#08H 
                MOV     DATAKIRIM,#$8 
                LJMP    TOLCD 
CTOMBOL0:       CJNE    A,#07H,KOLOM3 
                MOV     DATAKEY,#0AH 
                MOV     DATAKIRIM,#$0 
                LJMP    TOLCD 
KOLOM3:         SETB    X1 
                SETB    X2 
                CLR     X3 
                MOV     A,SWITCH 
                ANL     A,#0FH 
CTOMBOL3:       CJNE    A,#0EH,CTOMBOL6 
                MOV     DATAKEY,#03H 
                MOV     DATAKIRIM,#$3 
                LJMP    TOLCD 
CTOMBOL6:       CJNE    A,#0DH,CTOMBOL9 
                MOV     DATAKEY,#06H 
                MOV     DATAKIRIM,#$6 
                LJMP    TOLCD 
CTOMBOL9:       CJNE    A,#0BH,CTOMBOLP 
                MOV     DATAKEY,#09H 
                MOV     DATAKIRIM,#$9 
                LJMP    TOLCD 



CTOMBOLP:       CJNE    A,#07H,KEYPADRET 
                MOV     DATAKEY,#0CH 
                MOV     DATAKIRIM,#'P' 
TOLCD:       
                LCALL   DELAYSW 
                MOV     A,DATAKEY 
                RET 
KEYPADRET:      MOV     A,#$0 
                RET 
 
KEYPADTOLCD: 
                MOV     DPTR,#KEY 
                MOV     A,DATAKEY 
                MOVC    A,@A+DPTR 
                LCALL   WRITE_DATALCD 
                RET 
 
DELAY_LCD:      MOV     R7,#04H 
D_LCD1:         MOV     R6,#3FH         ;4F 
D_LCD2:         DJNZ    R6,D_LCD2 
                DJNZ    R7,D_LCD1 
                RET 
 
ANDF0:          SWAP    A 
AND0F: 
                ANL     A,#$0F 
                RET 
 
DELAYSW: 
                MOV     R5,#02H 
DELAYSW0:       MOV     R6,#0FFH 
DELAYSW1:       MOV     R7,#0FFH 
DELAYSW2:       DJNZ    R7,DELAYSW2 
                DJNZ    R6,DELAYSW1 
                DJNZ    R5,DELAYSW0 
                RET 
 
DELAY5X: 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
                LCALL   DELAYSW 
                RET 
 
 
TEXT1:          .BYTE           " KONTROL GEDUNG ",0 
TEXT2:          .BYTE           " - HANDY TALKY -",0 
TEXTPERINTAH1:  .BYTE           " M A S U K K A N",0 
TEXTPERINTAH2:  .BYTE           " PERINTAH= ",0 
KEY:            .BYTE           " 1234567890*#" 
 
 
                .END 
 
 
 



 
 
 
 

PROGRAM PADA BAGIAN SISTEM PLANT 
 

#INCLUDE        "8051.H" 
 
RELAY1          .EQU    P0.0 
RELAY2          .EQU    P0.1 
RELAY3          .EQU    P0.2 
RELAY4          .EQU    P0.3 
LEDSTATUS       .EQU    P1.0 
 
 
 
                .ORG    $30 
DATA            .BLOCK  1 
 
                .ORG    $0 
                LJMP    MULAI 
 
 
                .ORG    $100 
MULAI:          MOV     SP,#$20 
                MOV     PSW,#0 
                LCALL   DELAYLED 
                CLR     RELAY1 
                CLR     RELAY2 
                CLR     RELAY3 
                CLR     RELAY4 
                LCALL   INITSERIAL 
                CLR     LEDSTATUS 
                LCALL   DELAYLED 
                SETB    LEDSTATUS 
                LCALL   DELAYLED 
                CLR     LEDSTATUS 
                LCALL   DELAYLED 
                SETB    LEDSTATUS 
                LCALL   DELAYLED 
                CLR     LEDSTATUS 
                LCALL   DELAYLED 
                SETB    LEDSTATUS 
                LCALL   DELAYLED 
                LJMP    LOOP 
ISIKANDATA: 
                MOV     DATA,A 
                LJMP    TUNGGUDATA 
 
LOOP:           MOV     SP,#$20 
TUNGGUDATA:     CLR     RI 
                LCALL   GETCHR 
                CJNE    A,#$0D,ISIKANDATA 
                CLR     LEDSTATUS 
                LCALL   DELAYR 
                SETB    LEDSTATUS 



                MOV     A,DATA 
CEK1:           CJNE    A,#$1,CEK2 
                SETB    RELAY1 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
CEK2:           CJNE    A,#$2,CEK3 
                SETB    RELAY2 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
CEK3:           CJNE    A,#$3,CEK4 
                SETB    RELAY3 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
CEK4:           CJNE    A,#$4,CEK5 
                SETB    RELAY4 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
CEK5:           CJNE    A,#$5,CEK6 
                CLR     RELAY1 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
CEK6:           CJNE    A,#$6,CEK7 
                CLR     RELAY2 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
CEK7:           CJNE    A,#$7,CEK8 
                CLR     RELAY3 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
CEK8:           CJNE    A,#$8,CEKBINTANG 
                CLR     RELAY4 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
CEKBINTANG:     CJNE    A,#'B',CEKPAGAR 
                SETB    RELAY1 
                SETB    RELAY2 
                SETB    RELAY3 
                SETB    RELAY4 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
CEKPAGAR:       CJNE    A,#'P',PASSKODE 
                CLR     RELAY1 
                CLR     RELAY2 
                CLR     RELAY3 
                CLR     RELAY4 
                LCALL   NYALAKANLED 
                LJMP    LOOP 
 
PASSKODE:       CLR     RI 
                LJMP    LOOP 
 
 
 
INITSERIAL: 
                MOV     TMOD,#20H 
                MOV     TCON,#41H 
                MOV     TH1,#$D0 



                MOV     SCON,#50H 
                RET 
 
GETCHR:         JNB     RI,GETCHR 
                MOV     A,SBUF 
                ANL     A,#$7F 
                CLR     RI 
                RET 
 
NYALAKANLED:    CLR     LEDSTATUS 
                LCALL   DELAYLED 
                LCALL   DELAYLED 
                LCALL   DELAYLED 
                LCALL   DELAYLED 
                LCALL   DELAYLED 
                LCALL   DELAYLED 
                LCALL   DELAYLED 
                LCALL   DELAYLED 
                LCALL   DELAYLED 
                LCALL   DELAYLED 
                SETB    LEDSTATUS 
                CLR     RI 
                RET 
 
DELAYR: 
DELAYR0:        MOV     R6,#$0F 
DELAYR1:        MOV     R7,#$FF 
DELAYR2:        DJNZ    R7,DELAYR2 
                DJNZ    R6,DELAYR1 
                RET 
 
DELAYSW: 
                MOV     R5,#01H 
DELAYSW0:       MOV     R6,#0FFH 
DELAYSW1:       MOV     R7,#0FFH 
DELAYSW2:       DJNZ    R7,DELAYSW2 
                DJNZ    R6,DELAYSW1 
                DJNZ    R5,DELAYSW0 
                RET 
 
DELAYLED: 
                MOV     R5,#06H 
DELAYLED0:      MOV     R6,#0FFH 
DELAYLED1:      MOV     R7,#0FFH 
DELAYLED2:      DJNZ    R7,DELAYLED2 
                DJNZ    R6,DELAYLED1 
                DJNZ    R5,DELAYLED0 
                RET 
 
 
 
                .END 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

LAMPIRAN C 

DATASHEET 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
  
 
 
 
 



 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 











 
 
 
 



 
 
 







 
 

 
 
 
 
 



 
 

 
 
 
 
 



 

 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 
 
 



 

 
 
 
 
 
 
 



 

 
 
 
 
 



 
 
 
 
 
 



 
 
 
 
 
 



 
 
 
 
 
 
 



 
 
 
 
 
 

 



 
 
 
 
 
 
 



 



 



 
 
 
 
 
 
 
 
 
 
 



 
 
 
 
 



 
 
 
 
 
 



 
 
 
 
 
 
 
 



 



 



 



 
 
 
 
 
 
 
 
 
 
 



 

 
 
 
 
 



 
 
 
 
 
 
 
 
 
 
 
 



 
 
 
 
 
 
 



 
 
 
 
 
 
 













 


